

Microsoft® Money 日本語版 V1.1

2005/12/7

目次

概要	5
Microsoft Money 電子明細とは	5
Microsoft Money 電子明細 技術概要	5
Web サイトからのダウンロード	7
Microsoft Money へのダウンロード	7
ID の設定/HTML サンプル	8
Microsoft Money 電子明細ファイル	9
OFX について	9
口座種類について	9
タグについて	9
文字セットについて	10
文字数と日本語の関係について	10
日付及び時刻について	10
預金口座型ファイル作成例	13
OFX ヘッダーの作成	13
金融機関情報の作成	13
口座情報の作成	13
預金口座型明細情報の作成 <STMTTRNRS>	14
明細情報の書出し	15
残高情報の作成	16
預金口座型明細情報の終了	16
明細の種類 <TRNTYPE>	16
摘要 <NAME>	17
明細 ID <FITID>	18
明細 ID <FITID> の生成例	18
SIC コード	19
クレジットカード型ファイル作成例	21
サンプル	23
預金口座型 電子明細ファイル	23
クレジットカード型 電子明細ファイル	24
ASP サンプル	25

概要

Microsoft Money 電子明細とは

インターネット Web サイトを運営する金融機関は、Microsoft Money 電子明細技術を用いることで、Microsoft Money ユーザーに対して取引明細や残高などのさまざまな情報をダウンロードすることができます。

この技術を用いて、ユーザーは以下のことが行えます。

- 預金口座の明細を Microsoft Money に取り込むことができます。
- クレジットカードの利用明細を Microsoft Money に取り込むことができます。

Microsoft Money は、個人の持つすべての資産を一元管理できるため、通常はユーザーがこれらの明細を入力しますが、本技術の採用により入力が必要なくなるため、Microsoft Money ユーザーがこの技術をサポートする金融機関をより多く利用することが見込まれます。また、利用明細を電子的に保存できることで、通帳や郵送されてくる利用明細の代替手段として、紙媒体の削減にも繋がります。

Microsoft Money 電子明細 技術概要

Microsoft Money 電子明細技術とは、米国において Active Statement と呼ばれる技術を利用しています。これは、OFX 技術を基に、簡素化し、インターネット Web サイトから明細をダウンロードさせるために特化した技術です。

本文書では、Microsoft Money 電子明細ファイルの作成の方法と、それを Web サイトから Microsoft Money にダウンロードする方法について概説します。

Web サイトからのダウンロード

Microsoft Money へのダウンロード

Microsoft Money がインストールされているコンピュータでは、「.OFX」拡張子が定義されています。Microsoft Money 電子明細技術に基づくファイルがローカルなハードディスク上にある場合、Microsoft Money は以下のいずれかの方法によってこれを読み込み、Microsoft Money 明細帳に取り込むことができます。

- Explorer 上で、ファイルをダブルクリック、あるいはファイルを選択して、[ファイル] メニューから [開く] コマンドを実行する。この拡張子には、Microsoft Money の中に入っているプログラムが関連付けられ、この操作で Microsoft Money 明細帳に取り込むことができます。
- Microsoft Money の [ファイル] メニューから [インポート] コマンドを実行し、ファイルを選択する。

Web サイトからダウンロードする場合、Windows が「.OFX」拡張子のファイルであると認識すれば、以上の操作が起これ、Microsoft Money ファイル内に取り込むことができます。具体的には、以下のいずれか、あるいは両方の方法を採用します。

- Content-Type を「application/x-ofx」にすることで、Microsoft Money のインストールされているコンピュータはこれを電子明細ファイルであると認識し、Microsoft Money に渡します。

弊社 Active Server Page 技術の場合のサンプルは、25 ページを参照してください。以下のプログラムを最初に実行します。

```
Response.ContentType = "application/x-ofx"
```

- リンクの拡張子を「.OFX」にする。ダウンロード用のページの中に、<AHREF="/Scripts/Download.ofx">などとするので、Windows はダウンロードされる内容が電子明細ファイルであることを認識し、Microsoft Money に渡します。

リンクの拡張子を「.OFX」にした場合でも、MIME の Content-Type を設定することが推奨されます。

- また、Microsoft Money 日本語版と一部のブラウザの不具合を防ぐため、以下の HTTP ヘッダを書きこむことが推奨されます。「download」の部分は任意で構いません。Content-Disposition に対応する新しいブラウザであれば、ファイル名として表示されます。古いブラウザは、この HTTP ヘッダを無視します。

```
Content-Disposition: filename=download.ofx
```

ID の設定/HTML サンプル

ダウンロード用のリンクを設定する場合、そのリンクに「DownloadOFX」という ID を付けることで、将来のバージョンの Microsoft Money はよりスムーズに連携を行うことができます。

例えば、以下のような形で記述してください。

```
<A HREF="/Scripts/Download.asp" ID="DownloadOFX"><IMG  
SRC="Download.GIF"></A>
```

また、弊社提供の標準画像を利用する場合、以下のようになります。

```
<MAP NAME="MSMoney">  
  <AREA SHAPE="RECT" COORDS="235, 0, 289, 39"  
    HREF="Help URL" target="_blank">  
  <AREA SHAPE="RECT" COORDS="100, 0, 234, 39"  
    HREF="Download URL" ID="DownloadOFX">  
</MAP>  
<IMG align="bottom" src="MSMoney.gif" width="289"  
  usemap="#MSMoney" height="39" border="0">
```

標準画像を利用して上記 HTML を記述したサンプルページが弊社 Web サイト上の下記のページにあります。また、ヘルプ画面へのジャンプ先（上記「*Help URL*」）は、以下の URL を使用してください。

Help URL	http://www.microsoft.com/japan/products/money/OFX/help.htm
デモ用画面	http://www.microsoft.com/japan/products/money/OFX/demo/

ユーザーに一目で分かりやすくするためにも、また弊社がさまざまな場所で行う説明と合致させるためにも、標準画像の利用をお勧めします。標準画像は、デモ用画面から取り出してください。

Microsoft Money 電子明細ファイル

OFX について

概要でも述べたように、電子明細ファイルは、OFX 技術を基にしている、そのサブセットとして定義されています。

OFX、およびそのファイルフォーマットの仕様については、以下の Web サイト（英文）を参照してください。本仕様書では、日本における仕様においての注意点に対象を絞っており、OFX の仕様に関して基本的な理解があることを前提としています。OFX 仕様書を参照する場合、以下を参照してください。

OFX	http://www.ofx.net
-----	---

口座種類について

電子明細では、大きく 2 種類の口座をサポートします。

- 預金口座型
- クレジットカード明細

これらはそれぞれ Microsoft Money の資産の種類に対応します。電子明細ファイル作成者は、それぞれの事情に応じ、これらを適切に選択し、あるいは組み合わせてください。

タグについて

OFX は非常に広い範囲をカバーしており、電子明細はその中の一部の定義のみを使用します。

電子明細で利用できるタグは以下のもの、及び OFX において以下のタグ中で使用できると定義されているものに限られます。

SONRS	金融機関に関する情報を Microsoft Money に渡すために使用されます。
STMTTRNRS	預金口座型の明細をダウンロードする時に使用します。
CCSTMTTRNRS	クレジットカードの明細をダウンロードする時に使用します。

Microsoft Money のバージョンによっては、タグのデータを使用していない場合があります。調査で使った Microsoft Money がタグのデータを無視するような場合でも、電子明細に出力する場合は、OFX および電子明細の仕様に基づいたデータを出力するようにしてください。Microsoft Money のバージョンアップによって、従来使用していなかったタグに関して解釈するような仕様変更は十分に考えられます。

文字セットについて

OFX 1.0.2 仕様書は、ASCII に準じた文字のみを許しています。これを世界各国において使用可能にするため、本電子明細仕様書、及び OFX 1.5 仕様書は、UNICODE 2.0 で定められている UTF-8 文字セットを使用します。

UNICODE 及び UTF-8 については、書籍「The Unicode Standard, Version 2.0」及び、以下の Web サイトを参照してください。

Unicode Consortium	http://www.unicode.org
--------------------	---

また、これに合わせて、OFX ヘッダーも変更します。OFX ヘッダーについては、13 ページを参照してください。

なお、Windows API 以外で UNICODE 変換を行う場合、変換表の互換性に注意してください。オペレーティング システムによっては、異なる変換表を持っている場合があります。使用しているシステムに応じて正しい変換が行われるように注意してください。

文字数と日本語の関係について

本仕様書で「...文字以内」と表記されている部分は、日本語文字や ASCII 文字などの文字種に関わらず、字数で数えます。

日付及び時刻について

日付・時刻は、OFX 仕様書に定義される形で書き出します。日本の時刻であれば、以下の形式となります。

YYYYMMDDHHMMSS[+9:JST]

元のデータに例えば秒がない場合には、0 を埋めます。1997 年 12 月 15 日、14 時 35 分は例えば以下ようになります。

19971215143500[+9:JST]

最後の部分 ([+9:JST]) は、この時刻がグリニッジ標準時から 9 時間早い、JST タイムゾーンの時刻であることを表しています。

Microsoft Money は、電子明細を読み込む際に、ユーザーが使用している Windows のタイムゾーンの設定にしたがって、日時を変換します。

このため、ユーザーが海外で Money を使っている場合は、明細の日付が時差の影響を受ける場合があります。

こうした影響を避ける必要がある場合は、12:00:00 (GMT) を使用することを検討してください。この場合、一部の地域以外では、Windows のタイムゾーンの影響を受けたとしても同じ日付になります。

12:00:00 (GMT) を表記する場合は、以下のように書き出します。

YYYYMMDD120000

預金口座型ファイル作成例

ここでは預金口座型の電子明細ファイルを例に取り、順を追って作成方法を示します。

OFX ヘッダーの作成

まず OFX ヘッダーを作成します。

```
OFXHEADER:100
DATA:OFXSGML
VERSION:102
SECURITY:NONE
ENCODING:UTF-8
CHARSET:CSUNICODE
COMPRESSION:NONE
OLDFILEUID:NONE
NEWFILEUID:NONE
[空行]
```

金融機関情報の作成

OFX ヘッダの後には、金融機関情報を作成します。

```
<OFX>
<SIGNONMSGSRSV1>
  <SONRS>
    <STATUS>
      <CODE>0
      <SEVERITY>INFO
    </STATUS>
    <DTSERVER>[time of server]
```

[Time of server] は、サーバがこの情報を作り出した時刻を示します。時刻の書式は、10 ページを参照してください。

次に使用している言語と金融機関名を指定します。言語は、「JPN」が日本語を示します。

```
    <LANGUAGE>JPN
    <FI>
      <ORG>[bank name]
    </FI>
  </SONRS>
</SIGNONMSGSRSV1>
```

<ORG>タグで指定する金融機関名 [bank name] は、32 文字以内になるようにしてください。

以上を作成し、金融機関に関する情報は終わりです。

口座情報の作成

次に各口座に関する情報を書き出します。全体の構成としては、以下の形になります。

```
<BANKMSGSRSV1>
```

```

<STMTTRNRS>
    ...口座 1 に関する情報
</STMTTRNRS>
<STMTTRNRS>
    ...口座 2 に関する情報
</STMTTRNRS>
<STMTTRNRS>
    ...口座 3 に関する情報
</STMTTRNRS>
</BANKMSGSRSV1>
<CREDITCARDMSGSRSV1>
    <CCSTMTTRNRS>
        ...クレジットカード 1 に関する情報
    </CCSTMTTRNRS>
</CREDITCARDMSGSRSV1>
</OFX>

```

明細をダウンロードする口座の数だけ、**<STMTTRNRS>**ブロック（あるいは**<CCSTMTTRNRS>**ブロック）を続けます。一つの口座の明細をダウンロードする場合は、**<STMTTRNRS>**ブロックも一つで構いません。また、預金口座、あるいはクレジットカードのみの場合は、その外側のブロックごと省略してください。例えば、預金口座一つの場合、以下の形になります。

```

<BANKMSGSRSV1>
    <STMTTRNRS>
        ...口座 1 に関する情報
    </STMTTRNRS>
</OFX>

```

また、一つのファイルに預金口座とクレジットカードの両方を生成する場合、必ず預金口座情報をまず生成し、次にクレジットカードとしてください。添付のサンプルには、預金口座とクレジットカードそれぞれ一つずつのものがあ

預金口座型明細情報の作成 <STMTTRNRS>

<BANKMSGSRSV1>ブロック内で、以下の行を書き出すことで、預金口座型明細情報の開始を示します。

```

<STMTTRNRS>
    <TRNID>0
    <STATUS>
        <CODE>0
        <SEVERITY>INFO
    </STATUS>

```

<TRNID>は、双方向やり取りのある OFX トランザクションで使用されるものです。この場合には使用されないで、0 を埋めます。

次に使用する通貨を、ISO-4217 に定義されている 3 文字記号を使って書き出します。「JPY」が日本円を表します。

```

<STMTTRS>
    <CURDEF>JPY

```

次に口座に関する情報を書き出します。Microsoft® Money はこれを使って、

Microsoft® Money 内の口座資産と付き合わせを行います。

```
<BANKACCTFROM>
  <BANKID>[bank identifier]
  <BRANCHID>[branch identifier]
  <ACCTID>[account identifier]
  <ACCTTYPE>[account type]
</BANKACCTFROM>
```

[Bank identifier]、[branch identifier]、[account identifier]のそれぞれは、銀行番号、支店番号、口座番号をそれぞれ埋めます。すべて ASCII 文字列のみを使用し、それぞれ 9、22、22 字以内の文字列です。

次の[account type]は、口座の種類です。以下の表に従って、ANSER-SPC の口座の種類を書き込みます。

口座種類	ACCTTYPE
当座	CHECKING
その他	SAVINGS

明細情報の書出し

作成する明細期間の日付及び時刻を作成します。

```
<BANKTRANLIST>
  <DTSTART>[starting date/time]
  <DTEND>[ending date/time]
```

次に、一つ一つの明細についての情報を書き出します。<STMTTRN>と</STMTTRN>で囲まれたブロックが、明細の一行を表します。複数の明細がある場合には、このブロックを複数続けます。また、明細が一つもない場合には、このブロックを飛ばして、6 ページの残高から書き出します。

```
<STMTTRN>
  <TRNTYPE>[transaction type]
  <DTPOSTED>19951202155648[+9:JST]
  <TRNAMT>[amount]
  <FITID>[fitid]
  <NAME>[payee name]
  <MEMO>[memo]
</STMTTRN>
```

それぞれのタグの概略を以下に示します。より詳細な定義や最大文字列長、書式などは、OFX 仕様書を参照してください。

TRNTYPE	明細の種類を示します。16 ページを参照してください。
DTPOSTED	明細の時刻を示します。
TRNAMT	明細の金額を示します。正の数が口座への入金、負の数が出金を示します。
FITID	明細 ID を示します。255 字以内の文字列。18 ページを参照してください。
NAME	Microsoft Money 明細帳の摘要欄（支払先/支払元）に入れるべき

文字列を示します。32 字以内¹の文字列。摘要が不要な場合には、タグごと省略してください。また、摘要と備考の使い方については、17 ページを参照してください。

MEMO

Microsoft Money 明細帳の備考欄に入れるべき文字列を示します。255 字以内²の文字列。備考が不要な場合には、タグごと省略してください。

他にも SIC コード (19 ページ参照) など、OFX で定義されているタグを入れても構いません。

残高情報の作成

口座残高と、その残高を確認した日付・時刻を書き出します。

```
</BANKTRANLIST>
<LEDGERBAL>
  <BALAMT>[ledger balance amount]
  <DTASOF>[ledger balance date/time]
</LEDGERBAL>
```

預金口座型明細情報の終了

最後に、預金口座型明細情報の終わりを示すマーカを入れます。

```
</STMTRS>
</STMTRNRS>
```

これでファイルの作成は終了します。

明細の種類 <TRNTYPE>

TRNTYPE は、OFX 仕様書に従います。

以下に幾つかの解釈例を示します。

入出金	取引種類	TRNTYPE	制約条件
入金	利息	INT	TRNAMT > 0
	配当	DIV	TRNAMT > 0
	振込入金	DIRECTDEP	TRNAMT > 0
	取立入金	DIRECTDEP	TRNAMT > 0
	自動引落の戻し入金	DIRECTDEP	TRNAMT > 0
	その他	DEP	TRNAMT > 0

¹ Money 2004 以前のバージョンでは、Shift-JIS に変換後のバイト数が 32 バイトを超える場合、超える部分は無視されます。Money 2005 以降のバージョンでは、Shift-JIS への変換は行わず、文字数で処理します。

² Money2004 以前のバージョンでは、Shift-JIS に変換後のバイト数が 255 バイトを超える場合、超える部分は無視されます。Money 2005 以降のバージョンでは、Shift-JIS への変換は行わず、文字数で処理します。

入出金	取引種類	TRNTYPE	制約条件
出金	自動引落	PAYMENT	TRNAMT < 0
	振込	PAYMENT	TRNAMT < 0
	現金引出	CASH	TRNAMT < 0
	カードによる引出	ATM	TRNAMT < 0
	小切手関連取引	CHECK	TRNAMT < 0
	その他	DEBIT	TRNAMT < 0
その他		OTHER	なし

なお、幾つかの **TRNTYPE** には制約条件があり、それが満たされない場合、**Microsoft Money** はその **OFX** ファイルを読み取ることはできないことに注意してください。

これらの制約条件を満たさない場合、入出金のタイプに合わせた適切な **TRNTYPE** に置き換えてください。不明な場合には、**OTHER** とすることで、**TRNAMT** が正なら入金、負なら出金の扱いとなります。

摘要 <NAME>

Microsoft Money は、明細を読み込んだ後、それぞれの明細についてユーザーに費目を設定させることで、より正確なレポートを生成する機能を持っています。この機能を使用するには、<NAME>タグをサポートする必要があります。

この時、摘要の文字列が、過去の摘要文字列に一致すると、同じ費目設定を表示し、ユーザーに確認を求めることでユーザーの仕分けをより簡便にすることができます。このため、入出金の用途が同じ場合には、摘要の文字列はなるべく同じ文字列を使用することが望まれます。

例えば、カードによる振込の場合であれば、振込先口座名などを摘要とし、「カード振込」や、「内手数料 XX 円」などは備考とします。こうすることにより、同じ人に振り込む場合には、規定値として同じ費目が設定され、ユーザーの作業を軽減できます。

また、振込や時間外出金の場合のように手数料が付属する場合、手数料は別の明細にすることが推奨されます。これにより、ユーザーがより正確な費目分けを、より簡単に行うことができます。

摘要と備考の使い方について、幾つかの例を下に挙げます。

	摘要	備考
振込	振込先口座名義	「カード振込」、「内手数料 XX 円」など
引落	引落目的（電話代など）、または	

	摘要	備考
	引落口座名義	
手数料	金融機関名	「振込手数料」など
利息	金融機関名	「利息 XX 年 XX 月分」など
現金引出	「現金引出」、「カード引出」など	「カード引出」、「ATM」など

明細 ID <FITID>

明細 ID は、金融機関が設定する明細の ID です。「0」以外の、一つの口座の中で完全に固有になる、255 文字以下の文字列を使用してください。ASCII 文字のみ使用できます。

同一の明細が複数回ダウンロードされた場合、Microsoft Money は明細 ID を使ってその検出を行います。既に Microsoft Money 明細帳に入っている明細と同じ明細 ID を持つ明細がダウンロードされると、Microsoft Money はそれを検出し、消しこみを行います。このため、誤って同一の明細 ID が使用されると、ダウンロードされた明細が Microsoft Money の明細帳に表示されません。

サーバ側で明細 ID を作成できない場合のみ、FITID に「0」を使用することができます。この場合は、Microsoft Money は、FITID による明細の重複チェックができないため、Microsoft Money 独自の処理で重複をチェックし、必要に応じて消し込み等の処理を行います。この処理によって、一部の明細が読み込まれない場合があります。この処理は Money のバージョンに依存し、動作が変更になる場合があります。また、OFX の仕様では、FITID に「0」を指定することは許可されていないため、Microsoft Money 独自の処理になります。

明細 ID <FITID>の生成例

FITID は、金融機関が提供しているサービスで、すでに明細ごとのユニークな ID が取得できる場合は、その値を利用することができます。

こうした ID が取得できない場合に、すでに明細に含まれている情報から生成する例としては以下のものがあります。

明細の順序が入れ替わらない場合

銀行のように新しい取引が順番に追加され、その順序が変わらない場合は、FITID は以下のように生成できます。

<明細の日付><同日の取引の序数>

FITID	日時	支払先	金額
20051001_1	2005/10/1	ShopA	-3000
20051002_1	2005/10/2	ShopB	-2000
20051002_2	2005/10/2	給与	5000

明細の順序が入れ替わる場合

クレジットカードの場合のように明細の順序が変わる場合があります。

このような場合は、FITID に支払月を追加することで、FITID を生成できます。

<支払月><明細の日付><同日の取引の序数(確定分のみ)>

FITID	日時	支払先	金額	支払日
200512_20051001_1	2005/10/1	ShopA	-3000	2005/12
200511_20051001_1	2005/10/1	ShopB	-2000	2005/11
200512_20051002_1	2005/10/2	ShopB	-2000	2005/11

使用しているシステムや取引の表示順序などに例外が発生するような場合は、今回の方法だけでは実装できない場合があります。必要に応じて適切に実装する必要があります。

違う取引に同じ FITID が設定されていた場合は、Microsoft Money はすでに入力済みの同じ取引として処理し、取り込まれません。逆に、同じ取引に違う FITID が設定されていた場合は、別の取引として重複して取り込まれます。

SIC コード

<STMTTRN>ブロック内に、<SIC>タグを使って SIC コードを書き込むことができます。<SIC>タグを使用する場合は、<FITID>タグと<NAME>タグの間に挿入してください。

SIC コードを提供することで、Microsoft Money が読み込む際に自動的に適切な費目を割り振り、ユーザーの作業をより軽減することができます。

<SIC>タグは、OFX 仕様書にある通り、省略可能なので、すべての<STMTTRN>ブロックに存在しても、まったく存在しなくても、または幾つかの<STMTTRN>ブロックだけに存在しても構いません。

SIC コードのより詳細な情報に関しては、下記の Web ページを参照してください。

<http://www.naics.com/>

<http://www.webplaces.com/naics/>

Microsoft Money の SIC コードの解釈例を下記に示します。なお、この表は、解釈例であって、完全なリストではありません。

SIC コード	費目	費目内訳
5541, 5542	交通費	ガソリン代
5511, 5521, 5531-5533, 5599, 7523, 7524, 7531-7539, 7542, 7549	交通費	自動車維持費
4841	通信費	ケーブル TV
4911, 4931	水道光熱費	電気代
4953	住居費	ごみ処分料
4922-4925, 4932, 5983	水道光熱費	ガス代
4812-4815	通信費	電話電報代
4941, 4952	水道光熱費	上下水道代
5942	教育費	教科書や学用品
8229-8222, 8241, 8243, 8244, 8249, 8299	教育費	授業料
5811-5814	食費	外食
5912	保険医療費	医薬品
6051, 6760	支払手数料	銀行手数料
5712-5714, 5718, 5719, 5722, 5932	住居費	家具や電化製品

クレジットカード型ファイル作成例

クレジットカード口座は、幾つかのタグが異なるだけで、その構造のほとんどは預金口座型の場合と同様です。13 ページを参照してください。また、タグの違いについては、24 ページのサンプルを参照してください。

サンプル

UTF-8 で記述された日本語を文書中に記述することができないため、ここに示されたサンプルはすべて **ASCII** 文字列のみ使用しています。

日本語の入ったサンプルファイルは、添付の **BankJ001.OFX**、あるいは **CardJ001.OFX** を参照してください。

預金口座型 電子明細ファイル

```

OFXHEADER:100
DATA:OFXSGML
VERSION:102
SECURITY:NONE
ENCODING:UTF-8
CHARSET:CSUNICODE
COMPRESSION:NONE
OLDFILEUID:NONE
NEWFILEUID:NONE

<OFX>
<SIGNONMSGSRSV1>
  <SONRS>
    <STATUS>
      <CODE>0
      <SEVERITY>INFO
    </STATUS>
    <DTSERVER>20051115090000[+9:JST]
    <LANGUAGE>JPN
    <FI>
      <ORG>Woodgrove Bank
    </FI>
  </SONRS>
</SIGNONMSGSRSV1>
<BANKMSGSRSV1>
  <STMTRNRS>
    <TRNUID>0
    <STATUS>
      <CODE>0
      <SEVERITY>INFO
    </STATUS>
    <STMTRS>
      <CURDEF>JPY
      <BANKACCTFROM>
        <BANKID>1
        <BRANCHID>123
        <ACCTID>222-333-4446
        <ACCTTYPE>SAVINGS
      </BANKACCTFROM>
      <BANKTRANLIST>
        <DTSTART>20051114150000[+9:JST]
        <DTEND>20051114150000[+9:JST]
        <STMTRN>
          <TRNTYPE>ATM
          <DTPOSTED>20051114150000[+9:JST]
          <TRNAMT>-10000
          <FITID>15268694
          <NAME>Cash
          <MEMO>ATM transaction
        </TRNTYPE>
      </BANKTRANLIST>
    </STMTRS>
  </STMTRNRS>
</BANKMSGSRSV1>
</OFX>

```

```

        </STMTRN>
      </BANKTRANLIST>
    </LEDGERBAL>
    <BALAMT>12335678
    <DTASOF>20051115080000[+9:JST]
  </LEDGERBAL>
</STMTRS>
</STMTRNRS>
</BANKMSGSRV1>
</OFX>

```

クレジットカード型 電子明細ファイル

```

OFXHEADER:100
DATA:OFXSGML
VERSION:102
SECURITY:NONE
ENCODING:UTF-8
CHARSET:CSUNICODE
COMPRESSION:NONE
OLDFILEUID:NONE
NEWFILEUID:NONE

<OFX>
  <SIGNONMSGSRV1>
    <SONRS>
      <STATUS>
        <CODE>0
        <SEVERITY>INFO
      </STATUS>
      <DTSERVER>20050519121500[+9:JST]
      <LANGUAGE>JPN
      <FI>
        <ORG>a
      </FI>
    </SONRS>
  </SIGNONMSGSRV1>
  <CREDITCARDMSGSRV1>
    <CCSTMTRNRS>
      <TRNUID>0
      <STATUS>
        <CODE>0
        <SEVERITY>INFO
      </STATUS>
      <CCSTMTRS>
        <CURDEF>JPY
        <CCACCTFROM>
          <ACCTID>7777777-72
        </CCACCTFROM>
        <BANKTRANLIST>
          <DTSTART>20050519000000[+9:JST]
          <DTEND>20050519120000
          <STMTRN>
            <TRNTYPE>DEP

          <DTPOSTED>20050519120000

          <TRNAMT>23065
          <FITID>14703
          <NAME>Name1
          <MEMO>Memo1
        </STMTRN>
      </BANKTRANLIST>
    </CCSTMTRNRS>
  </CREDITCARDMSGSRV1>
</OFX>

```



```

<LEDGERBAL>
  <BALAMT>16543
  <DTASOF>20050519120000
</LEDGERBAL>
</CCSTMTRS>
</CCSTMTRNRS>
</CREDITCARDMSGSRSV1>
</OFX>

```

ASP サンプル

以下のサンプルは、Microsoft Access の MDB ファイルから明細情報を読み出し、電子明細ファイルを作成します。

本サンプルプログラムは以下の環境で実行を確認しました。

- Microsoft Windows NT® 4.0 SP3
- Microsoft Internet Information Server 4.0
- Microsoft Active Server Page
- Microsoft Data Access Component 2.x
- 適切なサンプルデータベースファイル

使用しているのシステムや環境に応じて適時変更してください。

```

<%@ LANGUAGE = "VBScript" %>
<%
Option Explicit

Const strRev = "$Revision: 12 $"

' -----
' CONFIGURATION
Const strFiName = "マイクロソフト銀行"
Const strDbPath = "¥¥kkdfd1¥¥public¥¥wwwroot¥¥AS¥¥ActvStmt.mdb"
Dim fDebug
fDebug = (Request.QueryString("DEBUG") <> "")

' -----
' Establish connection to database
Dim adoCon, adoRsAcct
Set adoCon = AdoConOpen

' -----
' Prepare HTTP header

Response.ExpiresAbsolute = 0 ' Disable client-side cache
Response.ContentType = "application/x-ofx"
Response.Addheader "Content-Disposition", "filename=download.ofx"

' -----
' MAIN PROGRAM - WRITE OFX HEADER AND CONTENTS

Dim strBuf

' Write OFX header
strBuf = "OFXHEADER:100" & vbNewLine & _
  "DATA:OFXSGML" & vbNewLine & _
  "VERSION:102" & vbNewLine & _

```

```

"SECURITY:NONE" & vbNewLine & _
"ENCODING:UTF-8" & vbNewLine & _
"CHARSET:CSUNICODE" & vbNewLine & _
"COMPRESSION:NONE" & vbNewLine & _
"OLDFILEUID:NONE" & vbNewLine & _
"NEWFILEUID:NONE" & vbNewLine & vbNewLine
Response.Write strBuf

' Write OFX tags
WriteOFX

' END MAIN PROGRAM
' -----

' -----
' DATABASE CONNECTIONS

Function AdoConOpen()
    Dim cn
    Set cn = Server.CreateObject("ADODB.Connection")
    cn.Provider = "Microsoft.Jet.OLEDB.3.51"
    cn.Properties("Data Source") = strDbPath
    cn.Open
    Set AdoConOpen = cn
End Function

Function StrSqlBuildCriteria(strSql, strCriteria)
    If strSql = "" Then strSql = " WHERE " Else strSql = strSql & " AND "
    StrSqlBuildCriteria = strSql & strCriteria
End Function

Function StrSqlBuildCriteriaEq(strSql, strKey, strValue)
    If strValue <> "" Then strSql = StrSqlBuildCriteria(strSql, strKey & " = " & strValue)
    StrSqlBuildCriteriaEq = strSql
End Function

Function AdoRsOpenAccount(strAcctTypeCriteria)
    Dim strSql
    strSql = StrSqlBuildCriteriaEq("", "UID", Request.QueryString("UID"))
    strSql = StrSqlBuildCriteriaEq(strSql, "ACCTKEY",
Request.QueryString("ACCTKEY"))
    If strAcctTypeCriteria <> "" Then strSql = StrSqlBuildCriteria(strSql, "ACCTTYPE" &
strAcctTypeCriteria)
    Set AdoRsOpenAccount = adoCon.Execute("SELECT * FROM Account" & strSql)
End Function

' -----
' OUTPUT STRING IN UTF-8 ENCODING
Dim objCodeConv
Sub WriteUTF8(s)
    If IsEmpty(objCodeConv) Then Set objCodeConv =
Server.CreateObject("Evita.Convert")
    Response.BinaryWrite objCodeConv.ToByteVec(objCodeConv.ToUTF8(s))
End Sub

' -----
' TAG-VALUE UTILITIES
Sub BlockBegin(s)
    Response.Write "<" & s & ">" & vbNewLine
End Sub

Sub BlockEnd(s)
    Response.Write "</" & s & ">" & vbNewLine

```

```

End Sub

Sub WriteTag(tag, strValue, fOptional, cchMax)
    ' Trim string to appropriate length if specified
    strValue = Trim(strValue)
    If cchMax > 0 And Not fDebug Then strValue = Trim(Left(strValue, cchMax))
    If IsNull(strValue) Or strValue = "" Then
        If fOptional Then Exit Sub
        AssertAbort "invalid record"
    End If

    ' Encode special characters like <, >, &
    strValue = Server.HtmlEncode(strValue)

    ' Write in UTF-8
    WriteUTF8 "<" & tag & ">" & strValue & vbNewLine
End Sub

' -----
' Format date in OFX format in JST
Function StrFormatOFXDate(date)
    StrFormatOFXDate = Year(date)
    StrFormatOFXDate = StrFormatOFXDate & Right("00" & Month(date), 2)
    StrFormatOFXDate = StrFormatOFXDate & Right("00" & Day(date), 2)
    StrFormatOFXDate = StrFormatOFXDate & Right("00" & Hour(date), 2)
    StrFormatOFXDate = StrFormatOFXDate & Right("00" & Minute(date), 2)
    StrFormatOFXDate = StrFormatOFXDate & Right("00" & Second(date), 2)
    StrFormatOFXDate = StrFormatOFXDate & "[+9:JST]"
End Function

' -----
' DEBUG UTILITIES
Sub AssertAbort(s)
    If fDebug Then
        ' Break into debugger
        STOP
    Else
        Response.Write "UNEXPECTED ERROR OCCURRED: " & s &
vbNewLine
        Response.End
    End If
End Sub

' -----
' WRITE OFX BLOCKS

Dim adoRsTrn

Sub WriteOFX()
    BlockBegin "OFX"
    WriteSignOnMsgsRs

    ' Write all banking accounts
    Set adoRsAcct = AdoRsOpenAccount("<> 'CREDITCARD'")
    If NOT adoRsAcct.EOF Then
        BlockBegin "BANKMSGSRSV1"
        Do While NOT adoRsAcct.EOF
            WriteStmtTrnRs
            adoRsAcct.MoveNext
        Loop
        BlockEnd "BANKMSGSRSV1"
    End If

```

```

' Write all credit card accounts
Set adoRsAcct = AdoRsOpenAccount("=" & "CREDITCARD")
If NOT adoRsAcct.EOF Then
    BlockBegin "CREDITCARDMSGSRVS1"
    Do While NOT adoRsAcct.EOF
        WriteCCStmtTrnRs
        adoRsAcct.MoveNext
    Loop
    BlockEnd "CREDITCARDMSGSRVS1"
End If

BlockEnd "OFX"
End Sub

Sub WriteSignOnMsgsRs
    BlockBegin "SIGNONMSGSRVS1"
    BlockBegin "SONRS"
    WriteStatusOK
    WriteTag "DTSERVER", StrFormatOFXDate(Now), False, 0
    WriteTag "LANGUAGE", "JPN", False, 0
    If strFiName <> "" Then
        BlockBegin "FI"
        WriteTag "ORG", strFiName, False, 32
        BlockEnd "FI"
    End If
    BlockEnd "SONRS"
    BlockEnd "SIGNONMSGSRVS1"
End Sub

Sub WriteStatusOK
    Response.Write "<STATUS><CODE>0<SEVERITY>INFO</STATUS>" &
vbNewLine
End Sub

' -----
' STMTTRNRS -- bank statement main block
Sub WriteStmtTrnRs
    BlockBegin "STMTTRNRS"
    WriteTag "TRNUID", 0, False, 0
    WriteStatusOK
    WriteStmtRs "STMTRS"
    BlockEnd "STMTTRNRS"
End Sub

' -----
' CCSTMTTRNRS -- credit card statement main block
Sub WriteCCStmtTrnRs
    BlockBegin "CCSTMTTRNRS"
    WriteTag "TRNUID", 0, False, 0
    WriteStatusOK
    WriteStmtRs "CCSTMTRS"
    BlockEnd "CCSTMTTRNRS"
End Sub

' -----
' STMTRS/CCSTMTRS -- bank/credit card statements
Sub WriteStmtRs(sSTMTRS)
    BlockBegin sSTMTRS
    WriteTag "CURDEF", "JPY", False, 0
    WriteAcctFrom
    Set adoRsTrn = adoCon.Execute("SELECT * FROM Trans WHERE ACCTKEY=" &
adoRsAcct("ACCTKEY") & " ORDER BY Date")
    If Not adoRsTrn.EOF Then

```

```

        BlockBegin "BANKTRANLIST"
        WriteTag "DTSTART", StrFormatOFXDate(adoRsTrn("Date")), False, 0
        WriteTag "DTEND", StrFormatOFXDate(Now), False, 0
        Do While NOT adoRsTrn.EOF
            WriteStmtTrn adoRsTrn("TRNTYPE"), adoRsTrn("Date"), _
                adoRsTrn("Amount"), adoRsTrn("FITID"), _
                adoRsTrn("NAME"), adoRsTrn("MEMO")
            adoRsTrn.MoveNext
        Loop
        BlockEnd "BANKTRANLIST"
    End If
    WriteBalance
    BlockEnd sSTMTRS
End Sub

' -----
' ACCTFROM -- bank/credit card account information
Function StrConcatWithDelimiter(str1, str2, strDelimiter)
    If str1 <> "" And str2 <> "" Then
        StrConcatWithDelimiter = str1 & strDelimiter & str2
    Else
        StrConcatWithDelimiter = str1 & str2
    End If
End Function

Sub WriteAcctFrom
    Dim strAcctType
    strAcctType = adoRsAcct("ACCTTYPE")
    Select Case strAcctType
        Case "CHECKING", "SAVINGS", "MONEYMRKT", "CREDITLINE"
            BlockBegin "BANKACCTFROM"
            WriteTag "BANKID", adoRsAcct("FIID"), False, 9
            WriteTag "BRANCHID", adoRsAcct("BRANCHID"), True, 22
            WriteTag "ACCTID", adoRsAcct("ACCTID"), False, 22
            WriteTag "ACCTTYPE", strAcctType, False, 0
            BlockEnd "BANKACCTFROM"
        Case "CREDITCARD"
            BlockBegin "CCACCTFROM"
            Dim s
            s = StrConcatWithDelimiter(adoRsAcct("FIID"), adoRsAcct("BRANCHID"),
            "-")
            s = StrConcatWithDelimiter(s, adoRsAcct("ACCTID"), "-")
            WriteTag "ACCTID", s, False, 22
            BlockEnd "CCACCTFROM"
        Case Else
            AssertAbort "invalid account type in the record"
    End Select
End Sub

Sub WriteStmtTrn(strTrnType, dtPosted, aTrnAmt, strFitId, strName, strMemo)
    ' TRNTYPE cannot be omitted, if not in database, fill in
    if IsEmpty(strTrnType) Or IsNull(strTrnType) Or strTrnType = "" Then strTrnType =
    "OTHER"
    BlockBegin "STMTRN"
    WriteTag "TRNTYPE", strTrnType, False, 0
    WriteTag "DTPOSTED", StrFormatOFXDate(dtPosted), False, 0
    WriteTag "TRNAMT", aTrnAmt, False, 0
    If IsNull(strFitId) Or strFitId = "" Then strFitId = 0
    WriteTag "FITID", strFitId, False, 255
    WriteTag "NAME", strName, True, 32
    WriteTag "MEMO", strMemo, True, 255
    BlockEnd "STMTRN"
End Sub

```

```

' -----
' LEDGARBAL -- bank/credit card balance
Sub WriteBalance
    BlockBegin "LEDGERBAL"
    WriteTag "BALAMT", adoRsAcct("BALANCE"), False, 0
    WriteTag "DTASOF", StrFormatOFXDate(adoRsAcct("BALASOF")), False, 0
    BlockEnd "LEDGERBAL"
    If Not IsNull(adoRsAcct("AVAIL")) Then
        BlockBegin "AVAILBAL"
        WriteTag "BALAMT", adoRsAcct("AVAIL"), False, 0
        WriteTag "DTASOF", StrFormatOFXDate(adoRsAcct("AVAILASOF")),
False, 0
        BlockEnd "AVAILBAL"
    End If
End Sub
%>

```